

Cast3M : la procédure PASAPAS et les procédures utilisateurs

François Di Paola

CEA Saclay,
<http://www-cast3m.cea.fr>

Automne 2024

Sommaire

- Rappels sur Cast3M et PASAPAS
- Fonctionnement de PASAPAS
- Procédures utilisateur
- **Solveur mécanique quasi-statique** - procédure UNPAS
 - exercice 1 : **force suiveuse**
 - exercice 2 : **rupture par suppression d'éléments**
- **Solveur thermique transitoire** - procédure TRANSNON
 - exercice 3 : **source de chaleur variable**
 - exercice 4 : contact **thermo mécanique**
- Annexes

Cast3M, quid ?

Logiciel de simulation utilisant la **méthode des éléments finis** en **mécanique/thermique** des **structures** et des **fluides**

Cast3M, quid ?




Logiciel de simulation utilisant la **méthode des éléments finis** en **mécanique/thermique** des **structures** et des **fluides**

- Résolution d'**équations aux dérivées partielles**

Cast3M, quid ?

Logiciel de simulation utilisant la **méthode des éléments finis** en **mécanique/thermique** des **structures** et des **fluides**




- Résolution d'**équations aux dérivées partielles**
- **Système complet** : solveur, pré/post-processeur, visualisation, import/export des données...



Cast3M, quid ?

Logiciel de simulation utilisant la **méthode des éléments finis** en **mécanique/thermique** des **structures** et des **fluides**

- Résolution d'**équations aux dérivées partielles**
- **Système complet** : solveur, pré/post-processeur, visualisation, import/export des données...




- Basé sur un langage de commande : **Gibiane** (orienté objet)


Nombreux domaines d'application

- **Mécanique des structures**


Quasi-statique (non linéarités matériau, géométrie, conditions limites)



Microstructure : agrégat polycristallin



Composant : liaison d'un tube




Batiment en béton armé (S. Durand)

Nombreux domaines d'application

- **Mécanique des structures**

Quasi-statique (non linéarités matériau, géométrie, conditions limites)

Contact/frottement, Flambage



Attachement d'un clip (contact + plasticité)


Nombreux domaines d'application

- **Mécanique des structures**

Quasi-statique (non linéarités matériau, géométrie, conditions limites)

Contact/frottement, **Flambage**

Dynamique (temporelle, modale, interaction fluide structure)



Premiers modes propres d'une cymbale

Nombreux domaines d'application


- **Mécanique des structures**

Quasi-statique (non linéarités matériau, géométrie, conditions limites)

Contact/frottement, **Flambage**

Dynamique (temporelle, modale, interaction fluide structure)

Rupture (XFEM, propagation dynamique, zones cohésives)



Rupture d'éprouvette CT, plasticité/endommagement, suppression d'éléments lors du calcul (S. Kebiri)

Nombreux domaines d'application

- **Mécanique des structures**

Quasi-statique (non linéarités matériau, géométrie, conditions limites)


Contact/frottement, **Flambage**

Dynamique (temporelle, modale, interaction fluide structure)

Rupture (XFEM, propagation dynamique, zones cohésives)

- **Thermique**

Conduction, convection, advection, rayonnement, changement de phase



Thermo mécanique d'un té de tuyauterie

Nombreux domaines d'application

- **Mécanique des structures**

Quasi-statique (non linéarités matériau, géométrie, conditions limites)

Contact/frottement, **Flambage**

Dynamique (temporelle, modale, interaction fluide structure)

Rupture (XFEM, propagation dynamique, zones cohésives)


- **Thermique**

Conduction, convection, advection, rayonnement, changement de phase

- **Mécanique des fluides**

- **Diffusion** multi espèces (loi de Fick)

- Fabrication additive, Métallurgie



Proportion de bainite lors d'une fabrication additive (C. Berthnier)

Nombreux domaines d'application

- **Mécanique des structures**

Quasi-statique (non linéarités matériau, géométrie, conditions limites)

Contact/frottement, **Flambage**

Dynamique (temporelle, modale, interaction fluide structure)

Rupture (XFEM, propagation dynamique, zones cohésives)


- **Thermique**

Conduction, convection, advection, rayonnement, changement de phase

- **Mécanique des fluides**

- **Diffusion** multi espèces (loi de Fick)

- Fabrication additive, Métallurgie



Simulation magnéto thermo hydrodynamique du soudage TIG (arc plasma + bain) (C. Nahed)

Nombreux domaines d'application

- **Mécanique des structures**

Quasi-statique (non linéarités matériau, géométrie, conditions limites)

Contact/frottement, Flambage

Dynamique (temporelle, modale, interaction fluide structure)

Rupture (XFEM, propagation dynamique, zones cohésives)

- **Thermique**

Conduction, convection, advection, rayonnement, changement de phase

- **Mécanique des fluides**


- **Diffusion** multi espèces (loi de Fick)

- Fabrication additive, Métallurgie

- Magnétostatique

- Couplage thermo-hygro-mécanique

- Optimisation topologique



Optimisation topologique d'un pont

Présentation de PASAPAS

- Objectif

Résolution de problèmes *non linéaires évolutifs* de manière incrémentale en **thermique** et en **mécanique**

le temps peut être physique (ex : thermique transitoire)
ou non (ex : plasticité avec chargement progressif)

→ on parle donc volontiers de *variable d'évolution*

- Types de non linéarités traitées

Comportement (plasticité, endommagement, matériaux variables, etc.)

Géométrie (grands déplacements)

Déformations (grandes rotations)

Conditions limites (rayonnement, frottement, pression suivieuse, etc.)

Utilisation de PASAPAS

- **Créer une table** contenant toutes les données du problème

```
tab1                                = TABL ;
tab1 . 'MODELE'                     = mod1 ET mod2 ;
tab1 . 'CARACTERISTIQUES'          = mat1 ET mat2 ;
tab1 . 'BLOCAGES_MECANIQUES'       = blo1 ;
tab1 . 'CHARGEMENT'                 = cha1 ET cha2 ET cha3 ;
tab1 . 'TEMPS_CALCULES'             = PROG 0.1 'PAS' 0.1 50. ;
tab1 . 'TEMPS_SAUVES'               = PROG 4. 8. 15. 16. 23. 42. ;
tab1 . 'PRECISION'                  = 1.E-6 ;
tab1 . 'GRANDS_DEPLACEMENTS'       = VRAI ;
```

- **Appeler la procédure**

```
PASAPAS tab1 ;
```

- **Post traitement** des résultats

Aperçu des paramètres d'entrée

● Généralités

'MODELE'	MMODEL	Équations à résoudre, formulation EF (MODE)
'CARACTERISTIQUES'	MCHAML	Paramètres matériau et/ou géométriques (MATE)
'CHARGEMENT'	CHARGE	Évolution des CL et chargements au cours du calcul (CHAR)

● Thermique

'BLOCAGES_THERMIQUES'	RIGIDITE	Matrice de blocage des CL de type Dirichlet (BLOQ, RELA)
'CELSIUS'	LOGIQUE	= VRAI si les températures sont en degrés Celsius
'TEMPERATURES'.0	CHPOINT	Conditions initiales

● Mécanique

'BLOCAGES_MECANIQUES'	RIGIDITE	Matrice de blocage des CL de type Dirichlet (BLOQ, RELA)
'GRANDS_DEPLACEMENTS'	LOGIQUE	Équilibre vérifié sur les configurations déformées
'DEPLACEMENTS'.0	CHPOINT	Conditions initiales
'CONTRAINTES'.0	MCHAML	Idem
'VARIABLES_INTERNES'.0	MCHAML	Idem
'DEFORMATIONS_INELASTIQUES'.0	MCHAML	Idem

● Mécanique (dynamique)

'DYNAMIQUE'	LOGIQUE	= VRAI si calcul dynamique
'AMORTISSEMENT'	RIGIDITE	Matrice d'amortissement
'VITESSES'.0	MCHAML	Conditions initiales
'ACCELERATIONS'.0	CHPOINT	Idem

● Instants de calcul et sauvegarde

'TEMPS_CALCULES'	LISTREEL	Liste des instants de calcul (PROG)
'TEMPS_SAUVES'	LISTREEL	Liste des instants pour lesquels les résultats sont conservés (PROG)

Aperçu des paramètres de sortie

- Résultats

'TEMPS'	TABLE	Instants de calcul, identiques aux 'TEMPS_SAUVES'
'TEMPERATURES'	TABLE	Champs solutions pour chaque 'TEMPS_SAUVES'
'PROPORTION_PHASE'	TABLE	Idem
'DEPLACEMENTS'	TABLE	Idem
'REACTIONS'	TABLE	Idem
'CONSTRAINTES'	TABLE	Idem
'DEFORMATIONS_INELASTIQUES'	TABLE	Idem
'VARIABLES_INTERNES'	TABLE	Idem
'VITESSES'	TABLE	Idem
'ACCELERATIONS'	TABLE	Idem

Exemples de post traitement

- Extraction des champs solution :
avec l'indice dans la table

```
sig1 = tab1 . 'CONTRAINTES' . 5 ;
```

ou bien avec l'instant de calcul

```
sig1 = PECHE tab1 'CONTRAINTES' 28.3 ;
```

- Tracé en mode graphique interactif :

```
EXPLORER tab1 ;
```

- Évolution temporelle d'un champ calculé :

```
ev1 = EVOL 'TEMP' tab1 'DEPLACEMENTS' 'UX' pt1 ;
```

Algorithme de PASAPAS

- Initialisations
- Boucle sur les pas de temps
 - Boucle de convergence thermo mécanique
 - Solveur thermique
 - Solveur mécanique
 - Convergence thermo mécanique ?
- Enregistrement des résultats
- Fin

Algorithme de PASAPAS

- Initialisations
- Boucle sur les pas de temps
 - Boucle de convergence thermo mécanique
 - Solveur thermique
 - Solveur mécanique
 - Convergence thermo mécanique ?
- Enregistrement des résultats
- Fin

⇒ Sous procédures (pour certaines tâches)

PAS_DEFA PAS_INIT PAS_ETAT PAS_MATE PAS_EPTH
PAS_MODL TRANSNON PAS_RAYO UNPAS PAS_RESU ...

Algorithme de PASAPAS

- Initialisations
- Boucle sur les pas de temps
 - Boucle de convergence thermo mécanique
 - Solveur thermique
 - Solveur mécanique
 - Convergence thermo mécanique ?
- Enregistrement des résultats
- Fin

⇒ Sous procédures (pour certaines tâches)


PAS_DEFA PAS_INIT PAS_ETAT PAS_MATE PAS_EPTH
PAS_MODL TRANSNON PAS_RAYO UNPAS PAS_RESU ...

⇒ Appels optionnels à des **procédures utilisateur**

PERSO1 PERSO2 REEV_MEC REEV_THE
CHARMECA CHARATHER PARATHER

ces procédures sont à créer par l'utilisateur !

Algorithme de PASAPAS



Accès aux données de PASAPAS

- `tab1.`'ESTIMATION'

contient les résultats calculés par **TRANSNON** et **UNPAS**

mais non convergées dans la **boucle de stationnarité thermo mécanique**

```
tab1 . 'ESTIMATION' . 'TEMPS '  
tab1 . 'ESTIMATION' . 'DEPLACEMENTS '  
tab1 . 'ESTIMATION' . 'CONTRAINTES '  
tab1 . 'ESTIMATION' . 'TEMPERATURES '  
etc...
```

- `tab1.`'CONTINUATION'

contient les résultats calculés et convergés (pour la **boucle thermo mécanique**)

cet indice est mis à jour à la fin du pas de temps !

utile pour une reprise de **PASAPAS**

```
tab1 . 'CONTINUATION' . 'TEMPS '  
tab1 . 'CONTINUATION' . 'DEPLACEMENTS '  
tab1 . 'CONTINUATION' . 'CONTRAINTES '  
tab1 . 'CONTINUATION' . 'TEMPERATURES '  
etc...
```

Accès aux données de PASAPAS

- `tab1.'WTABLE'`

Variables utiles à **PASAPAS** (options choisies, modèles, caractéristiques instanciées, chargements instanciés, résultats intermédiaires, ...)

Quelques indices :

`'WTABLE' . 'CHARGEMENT'`

Chargement

`'WTABLE' . 'THER_COURANT'`

Température à la dernière itération
(au cours d'un pas)

`'WTABLE' . 'BLOCAGES_MECANIQUES'`

Matrice de blocage mécanique

`'WTABLE' . 'BLOCAGES_THERMIQUES'`

Matrice de blocage thermique

`'WTABLE' . 'FOR'`

Configuration au début du pas

`'WTABLE' . 'FORO'`

Configuration initiale

`'WTABLE' . 'MODELE'`

Modèles

`'WTABLE' . 'CARACTERISTIQUES'`

Champ de caractéristiques matériau

etc. . .

Plus d'infos, voir les commentaires de la procédure **PAS_DEFA**

Procédures utilisateur

Mode d'emploi

Il existe 7 appels à des procédures utilisateurs
à différentes étapes de l'algorithme de **PASAPAS**

PERSO1 PERSO2 REEV_MEC REEV_THE
CHARMECA CHARATHER PARATHER

1. Choisir la procédure

selon son emplacement dans **PASAPAS** et l'action désirée
*par exemple : puisque PERSO1 est appelée après le calcul d'un pas de temps,
elle peut servir à modifier le suivant (conditions aux limites, matériau ...)*

2. Observer la syntaxe de la procédure

dans le code de **PASAPAS** ... ou bien voir le tableau suivant

Procédures utilisateur

Mode d'emploi

3. Définir la procédure

```
DEBP PERS01 t1*'TABLE' ;  
  MESS 'Bonjour tout le monde, PERS01 vous parle !' ;  
FINP;
```

4. Brancher la procédure dans PASAPAS

```
tab1 . 'MODELE' = mo ;  
tab1 . 'CARACTERISTIQUES' = ma ;  
tab1 . 'BLOCAGES_MECANIQUES' = bl ;  
tab1 . 'CHARGEMENT' = cha1 ;  
tab1 . 'TEMPS_CALCULES' = PROG 4. 8. 15. 16. 23. 42. ;  
tab1 . 'PROCEDURE_PERS01' = VRAI ;  
PASAPAS tab1 ;
```

Procédures utilisateur

Liste des procédures

Procédure	Indice dans la table PASAPAS	Syntaxe	Rôle possible
PERSO1	'PROCEDURE_PERSO1' = VRAI ;	PERSO1 tab1 ;	Mise à jour du problème après calcul du pas mécanique
REEV_MEC	'PROCEDURE_REEV_MEC' = VRAI ;	REEV_MEC tab1 n1 ;	Idem, mais dans la boucle thermo mécanique
CHARMECA	'PROCEDURE_CHARMECA' = VRAI ;	tab2 = CHARMECA tab1 tps1 ;	Ajout chargements mécaniques pendant UNPAS
PERSO2	'PROCEDURE_PERSO2' = VRAI ;	PERSO2 tab1 ;	Mise à jour du problème après calcul du pas thermique
REEV_THE	'PROCEDURE_REEV_THE' = VRAI ;	REEV_THE tab1 n1 ;	Idem, mais dans la boucle thermo mécanique
CHARTHER	'PROCEDURE_CHARTHER' = VRAI ;	tab2 = CHARTHER tab1 tps1 ;	Ajout chargements thermiques pendant TRANSNON
PARATHER	'PROCEDURE_PARATHER' = VRAI ;	PARATHER tab1 tps1 ;	Mise à jour variables externes des caractéristiques thermiques

Avec :

tab1 : la table de **PASAPAS**

n1 : le numéro d'appel de la procédure (0 ou 1)

tps1 : l'instant de calcul courant

tab2 : la table de sortie (pour **CHARMECA** et **CHARTHER**)

Procédures utilisateur

Quelques remarques

- Les instructions dans ces procédures sont libres !
- **CHARMECA** et **CHARTHER** doivent sortir une table, avec 2 indices possibles :
 - '**ADDI_MATRICE**' qui contient les matrices de RIGIDITE que l'on veut ajouter au 1er membre
 - '**ADDI_SECOND**' qui contient les CHPOINT que l'on veut ajouter au 2nd membre (forces nodales, ...)
- En grands déplacements (option '**GRANDS_DEPLACEMENTS**')
CHARMECA est appelée sur la configuration déformée

Procédures utilisateur


Exemple : éprouvette entaillée en traction

```
OPTI 'DIME' 2 'MODE' 'PLAN' 'DEFO' 'ELEM' 'TRI6' ;

OPTI 'DENS' 0.1 ;
p1 = 0. 0. ; p2 = 1. 0. ;
p3 = 1. 1. ; p4 = 0.55 1. ;
p7 = 0.45 1. ; p8 = 0. 1. ;
OPTI 'DENS' 0.005 ;
pc = 0.5 0.5 ;
p5 = pc PLUS ( 0.005 -0.0025) ;
p6 = pc PLUS (-0.005 -0.0025) ;
con = DROI p1 p2 DROI p3 DROI p4 DROI p5
      CERC pc p6 DROI p7 DROI p8 DROI p1 ;
msh = SURF con ;
l1 = con ELEM 'COMP' p8 p1 ;
l2 = con ELEM 'COMP' p2 p3 ;



mo = MODE msh 'MECANIQUE' ;
ma = MATE mo 'YOUN' 1.E9 'NU' 0.3 ;

b1 = (BLOQ 'UX' l1) ET (BLOQ 'UY' p1) ;
f1 = FSUR 'MASS' mo (1.E6 0.) l2 ;
ev1 = EVOL 'MANU' (PROG 0. 1.) (PROG 0. 1.) ;
cha1 = CHAR 'MECA' f1 ev1 ;
```



Procédures utilisateur

Exemple : éprouvette entaillée en traction



```

tab1                                     = TABL ;
tab1 . 'MODELE'                          = mo ;
tab1 . 'CARACTERISTIQUES'                = ma ;
tab1 . 'BLOCAGES_MECHANQUES'             = bl ;
tab1 . 'CHARGEMENT'                      = cha1 ;
tab1 . 'TEMPS_CALCULES'                   = PROG 0. 'PAS' 0.05 1. ;

```

```
PASAPAS tab1 ;
```

```

liso = PROG 0. 'PAS' 1.E6 20.E6 ;
REPE b1 (DIME tab1 . 'TEMPS') ;
  u1 = tab1 . 'DEPLACEMENTS' . (&b1 - 1) ;
  sig1 = tab1 . 'CONSTRAINTES' . (&b1 - 1) ;
  def1 = DEFO msh u1 10. ;
  TRAC sig1 mo def1 con liso ;
FIN b1 ;

```



Procédures utilisateur

Exemple : éprouvette entaillée en traction + flexion pilotée par l'ouverture

```
DEBP PERSO1 t1*'TABLE' ;
u = t1 . 'ESTIMATION' . 'DEPLACEMENTS' ;
du = (EXTR u 'UX' p4) - (EXTR u 'UX' p7) ;
SI ((t1 . 'PHASE_1') ET (du > 3.E-3)) ;
t1 . 'PHASE_1' = FAUX ;
tps = t1 . 'ESTIMATION' . 'TEMPS' ;
f2 = FSUR 'MASS' mo (0. -5.E5) l2 ;
ev2 = EVOL 'MANU' (PROG tps 1.) (PROG 0. 1.) ;
cha2 = CHAR 'MECA' f2 ev2 ;
t1 . 'WTABLE' . 'CHARGEMENT' = cha1 ET cha2 ;
FINSI ;
FINP ;

tab2 = TABL ;
tab2 . 'MODELE' = mo ;
tab2 . 'CARACTERISTIQUES' = ma ;
tab2 . 'BLOCAGES_MECANIQUES' = bl ;
tab2 . 'CHARGEMENT' = cha1 ;
tab2 . 'TEMPS_CALCULES' = PROG 0. 'PAS' 0.05 1. ;
tab2 . 'PROCEDURE_PERSO1' = VRAI ;
tab2 . 'PHASE_1' = VRAI ;
PASAPAS tab2 ;

liso = PROG 0. 'PAS' 1.E6 20.E6 ;
REPE b1 (DIME tab2 . 'TEMPS') ;
u1 = tab2 . 'DEPLACEMENTS' . (&b1 - 1) ;
sig1 = tab2 . 'CONSTRAINTES' . (&b1 - 1) ;
def1 = DEFO msh u1 10. ;
TRAC sig1 mo def1 con liso ;
FIN b1 ;
```



Solveur mécanique

Rappel des équations

Équations locales du problème statique

équilibre	$\text{div}(\boldsymbol{\sigma}) + \mathbf{f}_{\text{imp}} = 0$	sur ∂V
efforts imposés	$\boldsymbol{\sigma} \cdot \mathbf{n} = \mathbf{t}_{\text{imp}}$	sur ∂V_t
déplacements imposés	$\mathbf{u} = \mathbf{u}_{\text{imp}}$	sur ∂V_u

Solveur mécanique

Rappel des équations

Équations locales du problème statique

équilibre	$\text{div}(\boldsymbol{\sigma}) + \mathbf{f}_{\text{imp}} = 0$	sur ∂V
efforts imposés	$\boldsymbol{\sigma} \cdot \mathbf{n} = \mathbf{t}_{\text{imp}}$	sur ∂V_t
déplacements imposés	$\mathbf{u} = \mathbf{u}_{\text{imp}}$	sur ∂V_u

Formulation faible + discrétisation EF

$$\{F\}_{\text{ext}} - \int_V [B]^T \{\sigma\} dV = \{0\}$$

$$\underbrace{\int_{\partial V_t} [N]^T \{t\}_{\text{imp}} dS + \int_V [N]^T \{f\}_{\text{imp}} dV}_{\{F\}_{\text{imp}}} + \underbrace{\int_{\partial V_u} [N]^T \{\boldsymbol{\sigma} \cdot \mathbf{n}\} dS}_{\{F\}_{\text{reac}}} - \underbrace{\int_V [B]^T \{\sigma\} dV}_{\{F\}_{\text{int}} = [B]\{\sigma\}} = \{0\}$$

Solveur mécanique

Rappel des équations

Équations locales du problème statique

équilibre	$\text{div}(\boldsymbol{\sigma}) + \mathbf{f}_{\text{imp}} = 0$	sur ∂V
efforts imposés	$\boldsymbol{\sigma} \cdot \mathbf{n} = \mathbf{t}_{\text{imp}}$	sur ∂V_t
déplacements imposés	$\mathbf{u} = \mathbf{u}_{\text{imp}}$	sur ∂V_u

Formulation faible + discrétisation EF

$$\{F\}_{\text{ext}} - \int_V [B]^T \{\sigma\} dV = \{0\}$$

$$\underbrace{\int_{\partial V_t} [N]^T \{t\}_{\text{imp}} dS + \int_V [N]^T \{f\}_{\text{imp}} dV}_{\{F\}_{\text{imp}}} + \underbrace{\int_{\partial V_u} [N]^T \{\boldsymbol{\sigma} \cdot \mathbf{n}\} dS}_{\{F\}_{\text{reac}}} - \underbrace{\int_V [B]^T \{\sigma\} dV}_{\{F\}_{\text{int}} = [B]\{\sigma\}} = \{0\}$$

Vecteurs de forces nodales équivalentes :

$\{F\}_{\text{imp}}$	forces volumiques \mathbf{f}_{imp} et surfaciques \mathbf{t}_{imp} imposées
$\{F\}_{\text{reac}}$	réactions aux déplacements imposés \mathbf{u}_{imp}
$[B]\{\sigma\}$	forces volumique intérieures

PRES FSUR CNEQ
REAC
BSIG

Solveur mécanique

Résidu

- Résidu = mesure du déséquilibre

$$\{R\} = \{F\}_{\text{imp}} + \{F\}_{\text{reac}} - [B]\{\sigma\}$$

- Équilibre atteint lorsque

$$\|R\| < \varepsilon F_{\text{ref}}$$

$\|R\|$ norme du résidu (par exemple la norme infinie)
 ε précision du calcul (fournie par utilisateur)
 $\{F\}_{\text{ref}}$ effort de référence du problème considéré

Solveur mécanique

Comportement non linéaire

- Décomposition avec matrice de rigidité élastique linéaire

$$\sigma = \mathbb{E} : \varepsilon_{\text{lin}} + \sigma_{\text{nl}} \quad \text{avec} \quad \varepsilon_{\text{lin}} = \frac{1}{2}(\nabla u + \nabla^T u)$$
$$\{\varepsilon\}_{\text{lin}} = [B]\{U\}$$

Solveur mécanique

Comportement non linéaire

- Décomposition avec matrice de rigidité élastique linéaire

$$\sigma = \mathbb{E} : \varepsilon_{\text{lin}} + \sigma_{\text{nlin}} \quad \text{avec} \quad \varepsilon_{\text{lin}} = \frac{1}{2}(\nabla u + \nabla^T u)$$
$$\{\varepsilon\}_{\text{lin}} = [B]\{U\}$$

- L'équilibre s'écrit alors

$$[B]\{\sigma\} = \{F\}_{\text{imp}} + \{F\}_{\text{reac}}$$

$$[B][E]\{\varepsilon\}_{\text{lin}} = \{F\}_{\text{imp}} + \{F\}_{\text{reac}} - [B]\{\sigma\}_{\text{nlin}}$$

$$\underbrace{\int_V [B]^T [E] [B] dV}_{[K]} \{U\} = \{F\}_{\text{imp}} + \{F\}_{\text{reac}} - [B]\{\sigma\}_{\text{nlin}}$$

Solveur mécanique

Comportement non linéaire

- Décomposition avec matrice de rigidité élastique linéaire

$$\sigma = \mathbb{E} : \varepsilon_{\text{lin}} + \sigma_{\text{nlin}} \quad \text{avec} \quad \varepsilon_{\text{lin}} = \frac{1}{2}(\nabla u + \nabla^T u)$$

$$\{\varepsilon\}_{\text{lin}} = [B]\{U\}$$

- L'équilibre s'écrit alors

$$[B]\{\sigma\} = \{F\}_{\text{imp}} + \{F\}_{\text{reac}}$$

$$[B][E]\{\varepsilon\}_{\text{lin}} = \{F\}_{\text{imp}} + \{F\}_{\text{reac}} - [B]\{\sigma\}_{\text{nlin}}$$

$$\underbrace{\int_V [B]^T [E] [B] dV}_{[K]} \{U\} = \{F\}_{\text{imp}} + \{F\}_{\text{reac}} - [B]\{\sigma\}_{\text{nlin}}$$

$$[K]\{U\} = \{F\}_{\text{imp}} + \{F\}_{\text{reac}} - [B]\{\sigma\}_{\text{nlin}}$$

$\{U\}$	déplacements nodaux
$\{\varepsilon\}_{\text{lin}}$	déformations linéaires (petites déformations)
$\{\sigma\}_{\text{nlin}}$	contraintes non linéaires
$[E]$	matrice de Hooke
$[K]$	matrice de rigidité élastique

$$u(x) = [N(x)]\{U\}$$

$$\{\varepsilon\}_{\text{lin}} = [B]\{U\}$$

ELAS
RIGI

Solveur mécanique

Prise en compte des déplacements imposés

- Matrice de blocage

$$\begin{aligned} u &= u_{\text{imp}} \quad \text{sur} \quad \partial V_u \\ [A]\{U\} &= \{u\}_{\text{imp}} \end{aligned}$$

Solveur mécanique

Prise en compte des déplacements imposés

- Matrice de blocage

$$u = u_{\text{imp}} \quad \text{sur} \quad \partial V_u$$

$$[A]\{U\} = \{u\}_{\text{imp}}$$

- Multiplicateurs de Lagrange

depuis un problème linéaire et sans contrainte :
on ajoute les inconnues $\{\lambda\}$:

$$[K]\{U\} = \{F\}_{\text{imp}} - [B]\{\sigma\}_{\text{nlin}}$$

$$[K]\{U\} + \underbrace{[A]^T\{\lambda\}}_{-\{F\}_{\text{reac}}} = \{F\}_{\text{imp}} - [B]\{\sigma\}_{\text{nlin}} \quad \text{avec} \quad [A]\{U\} = \{u\}_{\text{imp}}$$

Solveur mécanique

Prise en compte des déplacements imposés

- Matrice de blocage

$$\begin{aligned} u &= u_{\text{imp}} \quad \text{sur} \quad \partial V_u \\ [A]\{U\} &= \{u\}_{\text{imp}} \end{aligned}$$

- Multiplicateurs de Lagrange

depuis un problème linéaire et sans contrainte :
on ajoute les inconnues $\{\lambda\}$:

$$[K]\{U\} = \{F\}_{\text{imp}} - [B]\{\sigma\}_{\text{nlín}}$$

$$[K]\{U\} + \underbrace{[A]^T\{\lambda\}}_{-\{F\}_{\text{reac}}} = \{F\}_{\text{imp}} - [B]\{\sigma\}_{\text{nlín}} \quad \text{avec} \quad [A]\{U\} = \{u\}_{\text{imp}}$$

L'équilibre s'écrit alors :

$$\underbrace{\begin{bmatrix} K & A^T \\ A & 0 \end{bmatrix}}_{[K]} \begin{Bmatrix} U \\ \lambda \end{Bmatrix} = \begin{Bmatrix} F_{\text{imp}} - [B]\sigma_{\text{nlín}} \\ u_{\text{imp}} \end{Bmatrix}$$

$[A]$	matrice blocages/rerelations	BLOQ RELA
$\{u\}_{\text{imp}}$	déplacements nodaux imposés	DEPI
$\{\lambda\}$	multiplicateurs de Lagrange	

Solveur mécanique

Prise en compte des déplacements imposés

- Matrice de blocage

$$\begin{aligned} u &= u_{\text{imp}} \quad \text{sur} \quad \partial V_u \\ [A]\{U\} &= \{u\}_{\text{imp}} \end{aligned}$$

- Multiplicateurs de Lagrange

depuis un problème linéaire et sans contrainte :
on ajoute les inconnues $\{\lambda\}$:

$$[K]\{U\} = \{F\}_{\text{imp}} - [B]\{\sigma\}_{\text{nlin}}$$

$$[K]\{U\} + \underbrace{[A]^T\{\lambda\}}_{-\{F\}_{\text{reac}}} = \{F\}_{\text{imp}} - [B]\{\sigma\}_{\text{nlin}} \quad \text{avec} \quad [A]\{U\} = \{u\}_{\text{imp}}$$

L'équilibre s'écrit alors :

$$\underbrace{\begin{bmatrix} K & A^T \\ A & 0 \end{bmatrix}}_{[K]} \begin{Bmatrix} U \\ \lambda \end{Bmatrix} = \begin{Bmatrix} F_{\text{imp}} - [B]\sigma_{\text{nlin}} \\ u_{\text{imp}} \end{Bmatrix}$$

$[A]$ matrice blocages/rerelations **BLOQ RELA**

$\{u\}_{\text{imp}}$ déplacements nodaux imposés **DEPI**

$\{\lambda\}$ multiplicateurs de Lagrange

$[K]$, $\{F\}_{\text{imp}}$, $[B]\{\sigma\}_{\text{nlin}}$ dépendents de $\{U\}$ (comportement non lin., grands dépl., forces suiveuses ...)

Solveur mécanique

Décomposition incrémentale 1/2

- Calcul d'un pas de temps entre t_0 et t_1

t_0 $\{U\}_0$ $\{\lambda\}_0$ $\{\sigma\}_0$

état connu

début du pas de temps

t_1 $\{U\}_1$ $\{\lambda\}_1$ $\{\sigma\}_1$

état recherché

fin du pas de temps

Solveur mécanique

Décomposition incrémentale 1/2

- Calcul d'un pas de temps entre t_0 et t_1
 - t_0 $\{U\}_0$ $\{\lambda\}_0$ $\{\sigma\}_0$ état connu début du pas de temps
 - t_1 $\{U\}_1$ $\{\lambda\}_1$ $\{\sigma\}_1$ état recherché fin du pas de temps
- Décomposition incrémentale des déplacements nodaux

$$\{U\}_1^i = \{U\}_0 + \Delta\{U\}_1^i \quad \text{et} \quad \Delta\{U\}_1^{i+1} = \Delta\{U\}_1^i + \delta\{U\}_1^{i+1}$$

et donc

$$\{U\}_1^{i+1} = \{U\}_1^i + \delta\{U\}_1^{i+1}$$

$\{U\}_1^i$
 $\Delta\{U\}_1^i$
 $\delta\{U\}_1^{i+1}$

estimation du déplacement $\{U\}_1^i$ à l'itération i
estimation de l'incrément de déplacement à l'itération i
correction de l'incrément de déplacement à l'itération i

Solveur mécanique

Décomposition incrémentale 1/2

- Calcul d'un pas de temps entre t_0 et t_1
 - t_0 $\{U\}_0$ $\{\lambda\}_0$ $\{\sigma\}_0$ état connu début du pas de temps
 - t_1 $\{U\}_1$ $\{\lambda\}_1$ $\{\sigma\}_1$ état recherché fin du pas de temps
- Décomposition incrémentale des déplacements nodaux

$$\{U\}_1^i = \{U\}_0 + \Delta\{U\}_1^i \quad \text{et} \quad \Delta\{U\}_1^{i+1} = \Delta\{U\}_1^i + \delta\{U\}_1^{i+1}$$

et donc

$$\{U\}_1^{i+1} = \{U\}_1^i + \delta\{U\}_1^{i+1}$$

$\{U\}_1^i$

estimation du déplacement $\{U\}_1$ à l'itération i

$\Delta\{U\}_1^i$

estimation de l'incrément de déplacement à l'itération i

$\delta\{U\}_1^{i+1}$

correction de l'incrément de déplacement à l'itération i

- Décomposition incrémentale des conditions sur les déplacements

$$[A]\{U\}_1^{i+1} = \{u\}_{\text{imp},1}$$

$$[A]\delta\{U\}_1^{i+1} = \{u\}_{\text{imp},1} - [A]\{U\}_1^i$$

Solveur mécanique

Décomposition incrémentale 2/2

- Décomposition incrémentale de l'équilibre

$$[K]\{U\}_1^{i+1} + [A]^T\{\lambda\}_1^{i+1} = \{F\}_{\text{imp},1} - [B]\{\sigma\}_{\text{nl},1}^i$$

$$[K]\delta\{U\}_1^{i+1} = \{F\}_{\text{imp},1} - [A]^T\{\lambda\}_1^{i+1} - \underbrace{([K]\{U\}_1^i + [B]\{\sigma\}_{\text{nl},1}^i)}_{[B]\{\sigma\}_1^i}$$

$$[K]\delta\{U\}_1^{i+1} = \underbrace{\{F\}_{\text{imp},1} - [A]^T\{\lambda\}_1^{i+1} - [B]\{\sigma\}_1^i}_{\{R\}_1^i}$$

Solveur mécanique

Décomposition incrémentale 2/2

- Décomposition incrémentale de l'équilibre

$$[K]\{U\}_1^{i+1} + [A]^T\{\lambda\}_1^{i+1} = \{F\}_{\text{imp},1} - [B]\{\sigma\}_{\text{nl},1}^i$$

$$[K]\delta\{U\}_1^{i+1} = \{F\}_{\text{imp},1} - [A]^T\{\lambda\}_1^{i+1} - \underbrace{([K]\{U\}_1^i + [B]\{\sigma\}_{\text{nl},1}^i)}_{[B]\{\sigma\}_1^i}$$

$$[K]\delta\{U\}_1^{i+1} = \underbrace{\{F\}_{\text{imp},1} - [A]^T\{\lambda\}_1^{i+1} - [B]\{\sigma\}_1^i}_{\{R\}_1^i}$$

L'équilibre s'écrit finalement

$$\underbrace{\begin{bmatrix} K & A^T \\ A & 0 \end{bmatrix}}_{[\hat{K}]} \begin{Bmatrix} \delta U_1^{i+1} \\ \lambda_1^{i+1} \end{Bmatrix} = \begin{Bmatrix} F_{\text{imp},1} - [B]\sigma_1^i \\ u_{\text{imp},1} - [A]U_1^i \end{Bmatrix}$$

Algorithme de minimisation du résidu

- Initialisations

$$[\{U\}_1^{i=0} \{\lambda\}_1^{i=0} \{\sigma\}_1^{i=0}] = [\{U\}_0 \{\lambda\}_0 \{\sigma\}_0]$$

$$\{F\}_{\text{reac},1}^{i=0} = -[A]^T \{\lambda\}_1^{i=0}$$

$$F_{\text{ref}} = \|\{F\}_{\text{imp},1} + \{F\}_{\text{reac},1}^{i=0}\|$$

$$\{R\}_1^{i=0} = \{F\}_{\text{imp},1} + \{F\}_{\text{reac},1}^{i=0} - [B]\{\sigma\}_1^{i=0}$$

initialisation de la solution

initialisation des réactions **REAC**

norme de convergence **MAXI 'ABS'**

premier résidu **BSIG**

Algorithme de minimisation du résidu

- Initialisations

$$[\{U\}_1^{i=0} \{\lambda\}_1^{i=0} \{\sigma\}_1^{i=0}] = [\{U\}_0 \{\lambda\}_0 \{\sigma\}_0]$$

$$\{F\}_{\text{reac},1}^{i=0} = -[A]^T \{\lambda\}_1^{i=0}$$

$$F_{\text{ref}} = \|\{F\}_{\text{imp},1} + \{F\}_{\text{reac},1}^{i=0}\|$$

$$\{R\}_1^{i=0} = \{F\}_{\text{imp},1} + \{F\}_{\text{reac},1}^{i=0} - [B]\{\sigma\}_1^{i=0}$$

initialisation de la solution

initialisation des réactions **REAC**

norme de convergence **MAXI 'ABS'**

premier résidu **BSIG**

- Tant que : $\|\{R\}_1^i\| \geq \varepsilon F_{\text{ref}}$

$$[\delta\{U\}_1^{i+1} \{\lambda\}_1^{i+1}] = [\hat{K}]^{-1}\{R\}_1^i$$

$$\{U\}_1^{i+1} = \{U\}_1^i + \delta\{U\}_1^{i+1}$$

$$\{\varepsilon\}_1^{i+1} = \mathcal{D}(\{U\}_1^{i+1})$$

$$\{\sigma\}_1^{i+1} = \mathcal{C}(\Delta\{\varepsilon\}_1^{i+1})$$

$$\{F\}_{\text{reac},1}^{i+1} = -[A]^T \{\lambda\}_1^{i+1}$$

$$\{R\}_1^{i+1} = \{F\}_{\text{imp},1} + \{F\}_{\text{reac},1}^{i+1} - [B]\{\sigma\}_1^{i+1}$$

résolution **RESO**

estim. déplacements

estim. déformations **EPSI**

estim. contraintes **COMP**


estim. réactions **REAC**

nouveau résidu **BSIG**



Fin


Algorithme de UNPAS



Exercice 1 : poutre avec force suiveuse

https://www-cast3m.cea.fr/index.php?page=exemples&exemple=formation_pasapas_1_initial


- Poutre en flexion
base encastrée, force **perpendiculaire** à la poutre, grands déplacements



Exercice 1 : poutre avec force suiveuse

https://www-cast3m.cea.fr/index.php?page=exemples&exemple=formation_pasapas_1_initial

- Poutre en flexion
base encastée, force **perpendiculaire** à la poutre, grands déplacements



- Problème : effort calculé sur la configuration initiale (non mis à jour)
- Objectif : mettre à jour la direction de l'effort au cours du calcul

👉 À vous de jouer !

Exercice 1 : poutre avec force suiveuse

https://www-cast3m.cea.fr/index.php?page=exemples&exemple=formation_pasapas_1_initial

- Quelques objets utiles
 - p2 point au sommet de la poutre, où est appliqué la force
 - ev1 évolution de l'amplitude de la force à appliquer vs. temps
- Quelques opérateurs utiles
 - EXTR** pour extraire des valeurs d'un champ
 - COS SIN** pour faire un peu de trigonométrie
 - IPOL** pour interpoler l'amplitude de la force à l'instant de calcul
 - FORC** pour appliquer une force ponctuelle
- Quelques indices utiles de la table
 - 'ESTIMATION'. 'DEPLACEMENTS' dernier champ de déplacement convergé
 - 'WTABLE'. 'CHARGEMENT' chargement courant

Exercice 1 : poutre avec force suivieuse

Solution avec PERSO1

- Utiliser la procédure **PERSO1** (1 appel / pas de temps) pour mettre à jour la force sur la configuration déformée au début du pas (explicite)
- Créer un objet CHARGEMENT et écraser 'WTABLE'. 'CHARGEMENT'

Programme principal

```
tab1 . 'PROCEDURE_PERSO1'   = VRAI ;
tab1 . 'MON_EVOL'           = ev1 ;
tab1 . 'MON_POINT'         = p2 ;
PASAPAS tab1 ;
```



Procédure PERSO1

```
DEBP PERSO1 t1*'TABLE' ;
  pf = t1 . 'MON_POINT' ;
  ev = t1 . 'MON_EVOL' ;
  * angle de rotation / rotation angle
  u1 = t1 . 'ESTIMATION' . 'DEPLACEMENTS' ;
  rs = EXTR u1 'RZ' pf ;
  rsd = rs * 180. / pi ;
  * direction force
  dir1 = (COS rsd) (SIN rsd) ;
  * écrasement du chargement / load overwrite
  f1 = FORC dir1 pf ;
  cha2 = CHAR 'MECA' f1 ev ;
  t1 . 'WTABLE' . 'CHARGEMENT' = cha2 ;
FINP ;
```

Exercice 1 : poutre avec force suivieuse

Solution avec PERSO1

- Résultats



Exercice 1 : poutre avec force suiveuse

Solution (bis) avec CHARMECA

- Idem mais avec la procédure **CHARMECA**, pour mettre à jour la force sur la configuration déformée au début du pas (explicite)
- Plus besoin de CHARGEMENT initial
- **Plus long (1 appel / itération / pas de temps)** et **résultats identiques**

Programme principal

```
***tab1 . 'CHARGEMENT'           = chai ;
tab1 . 'PROCEDURE_CHARMECA'     = VRAI ;
tab1 . 'MON_EVOL'               = ev1 ;
tab1 . 'MON_POINT'              = p2 ;
PASAPAS tab1 ;
```

Procédure CHARMECA

```
DEBP CHARMECA t1*'TABLE' tps1*'FLOTTANT' ;
  pf = t1 . 'MON_POINT' ;
  ev = t1 . 'MON_EVOL' ;
  * angle de rotation / rotation angle
  u1 = t1 . 'ESTIMATION' . 'DEPLACEMENTS' ;
  rs = EXTR u1 'RZ' pf ;
  rsd = rs * 180. / pi ;
  * direction force
  dir1 = (COS rsd) (SIN rsd) ;
  * ajout de la force / adding the force
  val1 = IPOL tps1 ev ;
  f1 = FORC (val1 * dir1) pf ;
  t2 = TABLE ;
  t2 . 'ADDI_SECOND' = f1 ;
FINP t2 ;
```

Exercice 1 : poutre avec force suiveuse

Solution (ter) avec CHARMECA

- Idem mais avec la procédure **CHARMECA**, pour mettre à jour la force sur la configuration déformée à la fin du pas (implicite)
- Attention : peut être instable !

Programme principal

```
***tab1 . 'CHARGEMENT'
tab1 . 'PROCEDURE_CHARMECA' = VRAI ;
tab1 . 'MON_EVOL'           = ev1 ;
tab1 . 'MON_POINT'         = p2 ;
PASAPAS tab1 ;
```


Procédure CHARMECA

```
DEBP CHARMECA t1*'TABLE' tps1*'FLOTTANT' ;
pf   = t1 . 'MON_POINT' ;
ev   = t1 . 'MON_EVOL' ;
* angle de rotation / rotation angle
drs = 0. ;
SI (EGA (TYPE zdept) 'CHPOINT') ;
    drs = EXTR zdept 'RZ' pf ;
FINSI ;
u1   = t1 . 'CONTINUATION' . 'DEPLACEMENTS' ;
rs   = (EXTR u1 'RZ' pf) + drs ;
rsd  = rs * 180. / pi ;
* direction force
dir1 = (COS rsd) (SIN rsd) ;
* ajout de la force / adding the force
val1 = IPOL tps1 ev ;
f1   = FORC (val1 * dir1) pf ;
t2   = TABLE ;
t2 . 'ADDI_SECOND' = f1 ;
FINP t2 ;
```

Exercice 2 : fissuration par suppression d'éléments

https://www-cast3m.cea.fr/index.php?page=exemples&exemple=formation_pasapas_2_initial


- Plaque en traction (déplacements imposés)



Exercice 2 : fissuration par suppression d'éléments

https://www-cast3m.cea.fr/index.php?page=exemples&exemple=formation_pasapas_2_initial

- Plaque en traction (déplacements imposés)



- Objectif : **supprimer les éléments** au cours du calcul
On utilisera un critère de rupture simple sur la 1^{ère} contrainte principale :

rupture si $\sigma_1 \geq 22 \text{ GPa}$

👉 À vous de jouer !

Exercice 2 : fissuration par suppression d'éléments

https://www-cast3m.cea.fr/index.php?page=exemples&exemple=formation_pasapas_2_initial

- Quelques opérateurs utiles
 - PRIN** calcul des contraintes principales
 - ELEM** sélection des éléments où un champ vérifie une condition
 - REDU** réduction d'un modèle sur un sous maillage
 - CHAN** changement des points support d'un champ
- Quelques indices utiles de la table
 - 'ESTIMATION'. 'CONSTRAINTES' dernier champ de contraintes convergé
 - 'WTABLE'. 'MODELE' modèle courant
 - 'WTABLE'. 'CARACTERISTIQUES' paramètres matériau

Exercice 2 : fissuration par suppression d'éléments

Solution avec PERSO1

- Utiliser la procédure **PERSO1**
- Extraire le modèle et les contraintes dans **'ESTIMATION'**
- Calculer la contrainte principale σ_I
- Extraire les éléments "intacts"
- Réduire le modèle sur ce maillage et écraser **'WTABLE' . 'MODELE'**
- Réduire les pas de temps

Programme principal

```
tab1 . 'TEMPS_CALCULES' =
  PROG 0.1 'PAS' 0.1 0.6 'PAS' 0.01 1. ;
tab1 . 'PROCEDURE_PERSO1' = VRAI ;
PASAPAS tab1 ;
```



Procédure PERSO1

```
DEBP PERSO1 t1*'TABLE' ;
  mo1 = t1 . 'WTABLE' . 'MODELE' ;
  sg1 = t1 . 'ESTIMATION' . 'CONSTRAINTES' ;
  sgp1 = PRIN sg1 mo1 ;
  sgi = EXCO 'SI11' sgp1 ;
  sg11 = CHAN 'GRAVITE' sgi mo1 ;
  s2 = sg11 ELEM 'INFERIEUR' 2.2E10 ;
  mo2 = REDU mo1 s2 ;
  t1 . 'WTABLE' . 'MODELE' = mo2 ;
FINP ;
```

Exercice 2 : fissuration par suppression d'éléments

Solution avec PERSO1

- Résultats



- Modèle peu robuste
résultats très sensibles à la discrétisation espace/temps

Exercice 2 : fissuration par suppression d'éléments

Solution (bis) avec PERSO1

- Idem mais sur les **blocages**
- Écraser 'WTABLE'. 'BLOCAGES_MECANIQUES'
- Activer l'option 'GRANDS_DEPLACEMENTS'

Programme principal

```
tab1 . 'TEMPS_CALCULES' =
  PROG 0.1 'PAS' 0.1 0.6 'PAS' 0.01 1. ;
tab1 . 'PROCEDURE_PERSO1' = VRAI ;
tab1 . 'GRANDS_DEPLACEMENTS' = VRAI ;
PASAPAS tab1 ;
```



Procédure PERSO1

```
DEBP PERSO1 t1*'TABLE' ;
mo1 = t1 . 'WTABLE' . 'MODELE' ;
mail1 = EXTR mo1 'MAIL' ;
sg1 = t1 . 'ESTIMATION' . 'CONTRAINTES' ;
sgp1 = PRIN sg1 mo1 ;
sgi = EXCO 'SI11' sgp1 ;
sg11 = CHAN 'GRAVITE' sgi mo1 ;
mail2 = sg11 ELEM 'INFERIEUR' 2.2E10 ;
mail3 = DIFF mail1 mail2 ;
SI ((NBEL mail3) > 0) ;
  b11 = t1 . 'WTABLE' . 'BLOCAGES_MECANIQUES' ;
  mb1 = EXTR b11 'MAIL' ;
  mb2 = mb1 ELEM 'APPU' 'LARG' mail3 ;
  mb3 = DIFF mb1 mb2 ;
  b12 = REDU b11 mb3 ;
  t1 . 'WTABLE' . 'BLOCAGES_MECANIQUES' = b12 ;
FINSI ;
FINP ;
```

Exercice 2 : fissuration par suppression d'éléments

Solution (bis) avec PERSO1

- Résultats



- Modèle peu robuste
résultats très sensibles à la discrétisation espace/temps

Solveur thermique

Rappel des équations

Équations locales du problème transitoire

équation de la chaleur $\rho c_p \frac{\partial T}{\partial t} + \operatorname{div} \underbrace{(-\lambda \operatorname{grad} T)}_{\phi} - q = 0$ sur V

flux imposé $\phi \cdot n = \phi_{\text{imp}} + \underbrace{h(T_f - T)}_{\text{convection}} + \underbrace{\varepsilon \sigma (T_{\infty}^4 - T^4)}_{\text{rayonnement}}$ sur ∂V_{ϕ}

température imposée $T = T_{\text{imp}}$ sur ∂V_T

Solveur thermique

Rappel des équations

Équations locales du problème transitoire

$$\text{équation de la chaleur} \quad \rho c_p \frac{\partial T}{\partial t} + \operatorname{div}(\underbrace{-\lambda \operatorname{grad} T}_{\phi}) - q = 0 \quad \text{sur } V$$

$$\text{flux imposé} \quad \phi \cdot n = \phi_{\text{imp}} + \underbrace{h(T_f - T)}_{\text{convection}} + \underbrace{\varepsilon \sigma (T_\infty^4 - T^4)}_{\text{rayonnement}} \quad \text{sur } \partial V_\phi$$

$$\text{température imposée} \quad T = T_{\text{imp}} \quad \text{sur } \partial V_T$$

Formulation faible et discrétisée :

$$[C]\{\dot{T}\} + [K]\{T\} = \{P\}$$

avec les matrices :

$$[C] = \int_V \rho c_p [N]^T [N] dV \quad \text{matrice de capacité} \quad \text{CAPA}$$

$$[K] = \int_V [B]^T [\lambda] [B] dV + \int_{\partial V_\phi} h [N]^T [N] dS \quad \text{matrice de conductivité} \quad \text{COND}$$

et le vecteur chargement nodal équivalent (SOUR FLUX CONV PAS_RAYO) :

$$\{P\} = \int_V [N]^T \{q\}_{\text{imp}} dV + \int_{\partial V_\phi} [N]^T (\{\phi\}_{\text{imp}} + h\{T\}_f + \varepsilon \sigma (\{T\}_\infty^4 - \{T\}^4)) dS$$

Solveur thermique

Discretisation temporelle avec un θ -schéma

$$[C]_{\star} \frac{\{T\}_1 - \{T\}_0}{\Delta t} + [K]_{\star} \{T\}_{\star} = \{P\}_{\star}$$

Avec :

- $\{T\}_1$ températures nodales à la fin du pas de temps t_1 (inconnues)
- $\{T\}_0$ températures nodales au début du pas de temps t_0 (connues)
- $\Delta t = t_1 - t_0$ le pas de temps


$[C]_{\star}$ $[K]_{\star}$ et $\{P\}_{\star}$ sont évalués :

- au temps $t_{\star} = \theta t_1 + (1 - \theta)t_0$
- à la température $\{T\}_{\star} = \theta \{T\}_1 + (1 - \theta)\{T\}_0$

θ est le **coefficient de relaxation** de ce schéma, entre 0 et 1

- θ réglable via l'indice 'RELAXATION.THETA'
- $\theta = 0$ schéma explicite
- $\theta = \frac{1}{2}$ schéma implicite de Crank-Nicolson
- $\theta = 1$ schéma implicite complet (par défaut)

Algorithme de TRANSNON



Lecture des options, unités (°C ou K), état initial **PAS_MODL**, calcul des facteurs de forme (rayonnement) **PAS_RAYO** ...

Mise à jour variables externes **PARATHER** et carac. matériaux, Calcul de $[K]_★$ **COND CONV**, $[C]_★$ **CAPA** à $\{T\}_0$ et $t_★$

Calcul de $\{P\}_★$ **TIRE** à $\{T\}_0$ et $t_★$

Ajout des termes de rayonnement **PAS_RAYO**

On définit alors l'opérateur $[L]_★ = \frac{1}{\Delta t}[C]_★ + \theta[K]_★$

Calcul incrément de température : $\Delta\{T\}_1^{i+1} = [L]_★^{-1}\{P\}_★$

on déduit $\{T\}_1^{i+1} = \{T\}_0 + \Delta\{T\}_1^{i+1}$

et aussi $\{T\}_★^{i+1} = \theta\{T\}_1^{i+1} + (1 - \theta)\{T\}_0$


Test sur la **variation d'incrément de température** entre 2 itérations successives

MAX 'ABS' $\frac{\Delta\{T\}_1^{i+1} - \Delta\{T\}_1^i}{\{T\}_1^{i+1}} < \text{'PRECISION'}$

Exercice 3 : dépendance à la température

https://www-cast3m.cea.fr/index.php?page=exemples&exemple=formation_pasapas_3_initial


- Section carrée chauffée par une source et refroidie par convection




Exercice 3 : dépendance à la température

https://www-cast3m.cea.fr/index.php?page=exemples&exemple=formation_pasapas_3_initial

- Section carrée chauffée par une source et refroidie par convection




- Objectif : rendre le problème **variable** !
 - Convection fonction du temps



Exercice 3 : dépendance à la température


https://www-cast3m.cea.fr/index.php?page=exemples&exemple=formation_pasapas_3_initial

- Section carrée chauffée par une source et refroidie par convection



- Objectif : rendre le problème **variable** !


- Convection fonction du temps
- Conductivité fonction de la température
 $\lambda(T) = 0.3T + 200$



Exercice 3 : dépendance à la température

https://www-cast3m.cea.fr/index.php?page=exemples&exemple=formation_pasapas_3_initial

- Section carrée chauffée par une source et refroidie par convection




- Objectif : rendre le problème **variable** !

- Convection fonction du temps
- Conductivité fonction de la température
 $\lambda(T) = 0.3T + 200$
- Source fonction de la température

$$q(T) = 4.10^6 \exp\left(-\left(\frac{T-1000}{700}\right)^2\right)$$

🎮 À vous de jouer !



Exercice 3 : dépendance à la température

https://www-cast3m.cea.fr/index.php?page=exemples&exemple=formation_pasapas_3_initial

- Quelques objets utiles
 - sou maillage de la source de chaleur
 - mosou modèle thermique réduit sur le maillage de la source
- Quelques opérateurs utiles
 - REDU** réduction des températures sur sou
 - SOUR** imposer une source volumique de chaleur
- Quelques indices utiles de la table
 - 'WTABLE'. 'THER_COURANT' températures courantes (itérations de **TRANSNON**)
 - 'WTABLE'. 'CHARGEMENT' chargement courant

Exercice 3 : dépendance à la température

Solution avec PERSO2

- Utiliser des objets EVOLUTION pour $\lambda(T)$ et $h(t)$
- Utiliser la procédure PERSO2
- Re-calculer la source selon la température
- Écraser 'WTABLE'. 'CHARGEMENT'

Programme principal

```
lt      = PROG 0. 1.E6 ;
lk      = (0.3 * lt) + 200. ;
evlam   = EVOL 'MANU' 'T' lt 'K' lk ;
macod   = MATE mocod 'K' evlam 'C' 128.

ltps    = PROG 0. 1.E5 ;
lh      = PROG 10. 200. ;
evh     = EVOL 'MANU' 'TEMP' ltps 'H' lh ;
macov   = MATE mocov 'H' evh 'TC' 20. ;

tab1 . 'PROCEDURE_PERSO2' = VRAI ;
tab1 . 'MON_MAIILLAGE'   = sou ;
TAB1 . 'MON_MODELE'     = mosou ;
PASAPAS tab1 ;
```

Procédure PERSO2

```
DEBP PERSO2 t1*'TABLE' ;
  masou = t1 . 'MON_MAIILLAGE' ;
  mosou = t1 . 'MON_MODELE' ;
  * reduction temperatures sur la source
  tp1   = t1 . 'WTABLE' . 'THER_COURANT' ;
  tsou  = REDU tp1 masou ;
  * calcul du champ de source selon temperature
  ex1   = ((tsou - 1000.) / 700.) ** 2 ;
  chs   = 4.E6 * (EXP (-1. * ex1)) ;
  * calcul du du second membre et chargement
  pmaj  = SOUR mosou chs ;
  chmaj = CHAR 'Q' pmaj ;
  * modification du chargement dans la table
  t1 . 'WTABLE' . 'CHARGEMENT' = chmaj ;
FINP ;
```

Exercice 3 : dépendance à la température

Solution avec CHARTHER

- Idem mais avec la procédure **CHARTHER**
- Pas besoin de CHARGEMENT initial

Programme principal

```

lt      = PROG 0. 1.E6 ;
lk      = (0.3 * lt) + 200. ;
evlam  = EVOL 'MANU' 'T' lt 'K' lk ;
macod  = MATE mocod 'K' evlam 'C' 128.

ltps   = PROG 0. 1.E5 ;
lh     = PROG 10. 200. ;
evh    = EVOL 'MANU' 'TEMP' ltps 'H' lh ;
macov  = MATE mocov 'H' evh 'TC' 20. ;

***tab1 . 'CHARGEMENT'          = chsou ;

tab1 . 'PROCEDURE_CHARTHER' = VRAI ;
tab1 . 'MON_MAIILLAGE'      = sou ;
tab1 . 'MON_MODELE'        = mosou ;
PASAPAS tab1 ;

```

Procédure CHARTHER


```

DEBP CHARTHER t1*'TABLE' tps1*'FLOTTANT' ;
masou = t1 . 'MON_MAIILLAGE' ;
mosou = t1 . 'MON_MODELE' ;
* reduction temperatures sur la source
tp1   = t1 . 'WTABLE' . 'THER_COURANT' ;
tsou  = REDU tp1 masou ;
* calcul du champ de source selon temperature
ex1   = ((tsou - 1000.) / 700.) ** 2 ;
chs   = 4.E6 * (EXP (-1. * ex1)) ;
* calcul du second membre
pmaj  = SOUR mosou chs ;
* sortie du second membre pour ajout
t2    = TABL ;
t2 . 'ADDI_SECOND' = pmaj ;
FINP t2 ;


```

Exercice 3 : dépendance à la température


- Résultats



Avec PERSO2




Avec CHARThER



Exercice 4 : contact thermo mécanique

https://www-cast3m.cea.fr/index.php?page=exemples&exemple=formation_pasapas_4_initial


- Dilatation thermique de 2 barreaux et mise en contact



Exercice 4 : contact thermo mécanique

https://www-cast3m.cea.fr/index.php?page=exemples&exemple=formation_pasapas_4_initial

- Dilatation thermique de 2 barreaux et mise en contact



- Objectif : ajouter le **contact thermique** entre 12 et 13, c'est à dire le transfert de chaleur lors du contact.

👉 À vous de jouer !

Exercice 4 : contact thermo mécanique

https://www-cast3m.cea.fr/index.php?page=exemples&exemple=formation_pasapas_4_initial

- Quelques objets utiles
p2 p3 points à gauche et à droite du jeu
12 13 lignes à gauche et droite du jeu
- Quelques opérateurs utiles
COOR obtenir les coordonnées de points
RELA imposer une relation linéaire entre degrés de liberté
- Quelques indices utiles de la table
'WTABLE' . 'BLOCAGES_THERMIQUES' CL thermiques courantes

Exercice 4 : contact thermo mécanique

Solution avec REEV_MEC

- Créer une **RELA**tion entre les températures de 12 et 13
- Utiliser la **boucle de convergence thermo mécanique**
- Utiliser la procédure **REEV_MEC**
- Calculer le jeu à l'itération courante et modifier (ou non) l'indice
'WTABLE' . 'BLOCAGES_THERMIQUES'

Programme principal

```
rel23 = RELA 'T' 12 - 'T' 13 ;
tab1 . 'CONVERGENCE_MEC_THE' = VRAI ;
tab1 . 'PROCEDURE_REEV_MEC' = VRAI ;
tab1 . 'POINT_2' = p2 ;
tab1 . 'POINT_3' = p3 ;
tab1 . 'RELA_L2_L3' = rel23 ;
PASAPAS tab1 ;
```



Procédure REEV_MEC

```
DEBP REEV_MEC t1*'TABLE' n1*'ENTIER' ;
  blt2 = t1 . 'BLOCAGES_THERMIQUES' ;
  rel1 = t1 . 'RELA_L2_L3' ;
  u1 = t1 . 'ESTIMATION' . 'DEPLACEMENTS' ;
  * calcul du jeu courant
  pt2 = t1 . 'POINT_2' ;
  pt3 = t1 . 'POINT_3' ;
  x2 = (COOR 1 pt2) + (EXTR u1 'UX' pt2) ;
  x3 = (COOR 1 pt3) + (EXTR u1 'UX' pt3) ;
  j1 = x3 - x2 ;
  * modification des blocages selon le jeu
  SI (j1 < 1.E-12) ;
    blt2 = blt2 ET rel1 ;
  FINSI ;
  t1 . 'WTABLE' . 'BLOCAGES_THERMIQUES' = blt2 ;
FINP ;
```

Exercice 4 : contact thermo mécanique

Solution avec REEV_MEC

● Résultats



Exercice 4 : contact thermo mécanique

Solution (bis) avec REEV_MEC

- Transfert par convection à l'interface avec des éléments de **RACC**ord coefficient d'échange dépendant de la pression
- Utiliser un objet **CHARGE**ment pour décrire la valeur de pression et mettre à jour ce chargement avec **REEV_MEC**

Programme principal

```
mrac = RACC (1.1 * j0) 12 13 ;
morac = MODE mrac 'THERMIQUE' 'CONVECTION' ;
hrac = EVOL 'MANU' 'PC' (PROG 0. 5.E8) 'H' (PROG 0. 1.E4) ;
marac = MATE morac 'H' hrac ;

cpc = MANU 'CHPO' mrac 'PC' 0. ;
cha3 = CHAR 'PC' cpc ;

tab1 . 'MODELE' = mom ET mot1 ET mot2 ET morac ;
tab1 . 'CARACTERISTIQUES' = mam ET mat1 ET mat2 ET marac ;

tab1 . 'CHARGEMENT' = cha1 ET cha2 ET cha3 ;

tab1 . 'CONVERGENCE_MEC_THE' = VRAI ;
tab1 . 'PROCEDURE_REEV_MEC' = VRAI ;
tab1 . 'LIGNE_REACTIONS' = 12 ;
tab1 . 'MALLAGE_RACCORD' = mrac ;
tab1 . 'CHARGEMENT_FIXE' = cha1 ET cha2 ;
PASAPAS tab1 ;
```

Exercice 4 : contact thermo mécanique

Solution (bis) avec REEV_MEC

- Transfert par convection à l'interface avec des éléments de **RACC**ord coefficient d'échange dépendant de la pression
- Utiliser un objet **CHARGE**ment pour décrire la valeur de pression et mettre à jour ce chargement avec **REEV_MEC**



Procédure REEV_MEC

```
DEBP REEV_MEC t1*'TABLE' n1*'ENTIER' ;
  r1  = t1 . 'ESTIMATION' . 'REACTIONS' ;
  lig = t1 . 'LIGNE_REACTIONS' ;
  mr  = t1 . 'MAILLAGE_RACCORD' ;
  cha0 = t1 . 'CHARGEMENT_FIXE' ;
  SI ((t1 . 'ESTIMATION' . 'TEMPS') NEG 0.) ;
* calcul de la pression de contact
  xpc = (MAXI 'ABS' (RESU (REDU r1 lig))) / ep ;
* chargement décrivant le parametre 'PC'
  cpc = MANU 'CHPO' mr 'PC' xpc ;
  cha3 = CHAR 'PC' cpc ;
* on écrase le chargement global de WTABLE
  t1 . 'WTABLE' . 'CHARGEMENT' = cha0 ET cha3 ;
  FINS ;
FINP ;
```

Exercice 4 : contact thermo mécanique

Solution (bis) avec REEV_MEC

● Résultats



Annexe : Paramètres de contrôle

● Généralités

'NB_BOTH'	ENTIER	Nombre max. d'itérations de la boucle de convergence thermo mécanique (10)
'MAXITERATION'	ENTIER	Nombre max. d'itérations (49)

● Mécanique

'PRECISION'	FLOTTANT	Critère pour comparer le résidu (1.E-4)
'FTOL'	FLOTTANT	Tolérance pour l'équilibre des efforts
'MTOL'	FLOTTANT	Tolérance pour l'équilibre des moments
'GRANDS_DEPLACEMENTS'	LOGIQUE	Configuration de référence = configuration déformée (FAUX)
'PREDICTEUR'	MOT	= 'HPP' Calcul en hypothèse "petits" déplacements → 1ère convergence puis passage grands déplacements → 2ème convergence
'PRECISINTER'	FLOTTANT	Précision pour le problème d'intégration des lois constitutives (1.E-8)
'CONVERGENCE_FORCEE'	LOGIQUE	Utilisation, ou non, de la convergence forcée (VRAI)
'MAXSOUSPAS'	ENTIER	Nombre max. de sous pas en convergence forcée (100)
'DELTAITER'	ENTIER	Nombre de pas d'écart pour test de non convergence (10)

Annexe : Paramètres de contrôle

- Thermique

'PROCEDURE_THERMIQUE'	MOT	Procédure de calcul à utiliser : = 'NONLINEAIRE' procédure TRANSNON (défaut) = 'LINEAIRE' procédure TRANSLIN = 'DUPONT' procédure DUPONT2
'RELAXATION_THETA'	FLOTTANT	Coefficient de relaxation pour la θ -méthode (1)

- Couplage thermo mécanique

'CONVERGENCE_MEC_THE'	LOGIQUE	Indique que l'on souhaite ré itérer la boucle de calcul thermo mécanique en cas de dépendance mutuelle (FAUX)
'CRITERE_COHERENCE'	FLOTTANT	Précision pour la convergence thermo mécanique, test sur la thermique (= 'PRECISION')
'PROJECTION'	LOGIQUE	Indique que le problème est couplé mais que les maillages en mécanique et en thermique sont différents (FAUX)

Annexe : Minimisation du résidu en Gibiane

- Algorithme **UNPAS** simplifié
→ voir la procédure **@SOLVMEC**

```
fex1 = TIRE char tps1 'MECA' ;  
flx1 = TIRE char tps1 'DIMP' ;  
res0 = fex1 + fr0 - fi0 + flx1 - flx0 ;  
fref = MAXI 'ABS' (fex1 + fr0) ;
```

```
k      = (RIGI mod mat) ET blo ;  
REPE b1 100 ;  
    ddu = RESO k res0 ;  
    u1  = u0 + ddu ;  
    fr1 = REAC blo u1 ;  
  
    eps1 = EPSI mod u1 ;  
  
    sig1 = ... COMP eps0 eps1 ... ;
```

```
    fi1 = BSIG mod sig1 ;  
  
    res1 = fex1 + fr1 - fi1 ;  
    cri1 = (MAXI 'ABS' res1) / fref ;  
    SI (cri1 < precis) ;  
        QUIT b1 ;  
    FINSI ;  
    u0 = u1 ;  
    res0 = res1 ;  
FIN b1 ;
```

Annexe : Minimisation du résidu en Gibiane

- Algorithme **UNPAS** simplifié
→ voir la procédure **@SOLVMEC**
grand déplacements

```
fex1 = TIRE char tps1 'MECA' ;
flx1 = TIRE char tps1 'DIMP' ;
res0 = fex1 + fr0 - fi0 + flx1 - flx0 ;
fref = MAXI 'ABS' (fex1 + fr0) ;
cof0 = FORM u0 ;
u00 = u0 ;
k = (RIGI mod mat) ET blo ET (KSIG mod sig0) ;
REPE b1 100 ;
  ddu = RESO k res0 ;
  u1 = u0 + ddu ;
  fr1 = REAC blo u1 ;
  du = u1 - u00 ;
  deps = EPSI mod du ;
  eps1 = eps0 + deps ;
  sig1 = ... COMP eps0 eps1 ... ;
  sig1 = PICA mod sig1 du ;
  FORM u1 ;
  fi1 = BSIG mod sig1 ;
  FORM cof0 ;
  res1 = fex1 + fr1 - fi1 ;
  cri1 = (MAXI 'ABS' res1) / fref ;
  SI (cri1 < precis) ;
    QUIT b1 ;
  FINSI ;
  u0 = u1 ;
  res0 = res1 ;
FIN b1 ;
```

Annexe : Critères de convergence (UNPAS)

- Normes de convergence (après le 1er **RESO**)

$$F_{\text{ref}} = \frac{\left| \delta\{U\}_1^1 \cdot \{F\}_{\text{imp}} - \{\lambda\}_1^1 \cdot (\{u\}_{\text{imp}} - [A]^T \cdot \{U\}_1^0) \right|}{\max_{\text{ddl depl}} \left| \delta\{U\}_1^1 \right| + \text{xpetit}} + \frac{\max_{\text{ddl depl}} \left| \delta\{U\}_1^1 \right| \cdot \max \left| [B] \{ \sigma \}_1^0 \right|}{\max_{\text{ddl depl}} \left| \delta\{U\}_1^1 \right| + \text{xpetit}} + \max \left| [B] \{ \sigma \}_1^0 \right|$$

$$M_{\text{ref}} = \frac{\left| \delta\{U\}_1^1 \cdot \{F\}_{\text{imp}} - \{\lambda\}_1^1 \cdot (\{u\}_{\text{imp}} - [A]^T \cdot \{U\}_1^0) \right|}{\max_{\text{ddl rota}} \left| \delta\{U\}_1^1 \right| + \text{xpetit}} + \frac{\max_{\text{ddl depl}} \left| \delta\{U\}_1^1 \right| \cdot \max \left| [B] \{ \sigma \}_1^0 \right|}{\max_{\text{ddl rota}} \left| \delta\{U\}_1^1 \right| + \text{xpetit}} + \text{xpetit}$$

- Mesures du résidu (à chaque itération de **UNPAS**)

$$x_{\text{conv}} = \frac{\max \left\{ \max_{\text{ddl depl}} \left| \{F\}_{\text{imp}} - [A]^T \cdot \{\lambda\}_1^{i+1} - [B] \cdot \{\sigma\}_1^{i+1} \right| ; \max \left| [A]^T (\{\lambda\}_1^{i+1} - \{\lambda\}_1^i) \right| \right\}}{F_{\text{ref}}}$$

$$x_{\text{convm}} = \frac{\max_{\text{ddl rota}} \left| \{F\}_{\text{imp}} - [A]^T \cdot \{\lambda\}_1^{i+1} - [B] \cdot \{\sigma\}_1^{i+1} \right|}{M_{\text{ref}}}$$

Annexe : Critères de convergence (UNPAS)

- Mesure de la variation d'incrément de déformation entre 2 itérations

$$\text{depstdm} = \left| \Delta\{\varepsilon\}_1^{i+1} - \Delta\{\varepsilon\}_1^i \right|$$

- La convergence est atteinte si :

$$\begin{aligned} \text{xconv} &< \text{'PRECISION'} \\ \text{xconvm} &< \text{'PRECISION'} \\ \text{depstdm} &< \text{'PRECISION'} \end{aligned}$$